Using SAS Proc IML to Generate Coefficients for Orthogonal Polynomials – Equal and Non-equal Spacing and Sample Sizes
General:  Orthogonal contrasts are used frequently in statistical analyses when treatments are structured in a manner designed to evaluate a response surface.  When treatments are equally spaced and the number of observations per treatment is balanced, tables are available in standard statistical textbooks to determine the coefficients for linear, quadratic, cubic, etc. contrasts.  Treatment spacing that is unequal is used frequently in experiments;  however, coefficients for unequally spaced contrasts are not readily available.  In addition, unbalanced sample sizes will affect computation of orthogonal contrasts.  Fortunately, Proc IML of SAS can be used to generate coefficients for both unequally spaced contrasts and unequal numbers of observations per treatment.  Two examples are provided here to illustrate the use of Proc IML using the ORPOL procedure;  one for equally spaced treatments and one for unequally spaced treatments.  The weight option of the ORPOL procedure can be used in either case to adjust for differences in sample size among treatment groups.
Example 1 – Equal Spacing and Balanced Numbers of Observations per Treatment:  The SAS code for generating the coefficient matrix using Proc IML follows.  This example is for four equally spaced treatments (e.g., 0, 1, 2, and 3% of a dietary additive).  For SAS, this code should be input to the Program Editor window and the job submitted with the F8 (run) key.

proc iml;

x={0 1 2 3}; /*Equal treatment spacing*/
xp=orpol(x,3);

print xp[label="Equal Treatment Spacing and Equal Treatment Sample Size"];
run; quit;

Note that the parenthetical expression in the third line contains the number three (x, 3).  The value 3 here represents the degrees of freedom for treatment (four treatments minus one = three degrees of freedom).  Change this value as needed for the number of treatments in each specific case.

The printed output from the execution of the statements above is shown below.  The second through the fourth columns contain the coefficients for the linear, quadratic, and cubic contrasts.

	First Column
	Linear Coefficients
	Quadratic Coefficients
	Cubic Coefficients

	0.5
	-0.67082
	0.5
	-0.223607

	0.5
	-0.223607
	-0.5
	0.6708204

	0.5
	0.2236068
	-0.5
	-0.67082

	0.5
	0.6708204
	0.5
	0.2236068


These coefficients can be used directly;  however, it can be readily determined that the linear coefficients can be converted (by scaling) to –3, –1, 1, and 3.  Similarly, the quadratic coefficients can be scaled to 1, –1, –1, and 1, whereas the cubic coefficients can be scaled to –1, 3, –3, and 1.  These scaled coefficients correspond to those that can be obtained from standard statistical textbooks.

Example 2 – Unequal Spacing and Unbalanced Numbers of Observations per Treatment:  The SAS code for generating the coefficient matrix using Proc IML follows.  This example is for four unequally spaced treatments (e.g., 0, 1, 2, and 4% of a dietary additive).  As before, in SAS, this code should be input to the Program Editor window and the job submitted with the F8 key.

proc iml;

x1={0 1 2 4}; /*Unequal treatment spacing and unequal sample size*/
w={5 4 5 4}; /*Sample size weights for treatments*/
x1p=orpol(x1,3,w);

print x1p[label="Unequal Treatment Spacing Weighted by Sample Size"];
run; quit;

The printed output from the execution of the statements above is shown below.  As before, the second through the fourth columns contain the coefficients for the linear, quadratic, and cubic contrasts.

	First Column
	Linear Coefficients
	Quadratic Coefficients
	Cubic Coefficients

	0.2357023
	-0.270369
	0.2389301
	-0.119404

	0.2357023
	-0.108148
	-0.155989
	0.3980149

	0.2357023
	0.0540738
	-0.290673
	-0.238809

	0.2357023
	0.3785166
	0.2206679
	0.0497519


As in the first example, the coefficients can be scaled;  however, it may be just as convenient to use the coefficients directly without scaling.

SAS Code for Contrast Statements:  To use the coefficients obtained from Proc IML for Example 2, the following SAS code should be added after the Proc GLM or Proc Mixed model statement.  This example assumes that the treatment variable is coded in the input statement as ‘trt’ and that the design is completely random.

proc glm;  classes trt;

model variables = trt/ss3;

lsmeans trt/stderr;

contrast ‘linear’ trt -0.270369 -0.108148 0.0540738 0.3785166;

contrast ‘quad’ trt 0.2389301 -0.155989 -0.290673 0.2206679;

contrast ‘cubic’ trt -0.119404 0.3980149 -0.238809 0.0497519;

run; quit;

As always, one needs to ensure that coefficients are ordered in the same way treatments are ordered.  It is a good idea to check the SAS output of the LSMEANS to determine the order of treatments or to alphabetically order the treatments in the input (e.g., in this example, treatment coding could be a_0, b_1, c_2, and d_4 to represent the 0, 1, 2, and 4% treatment levels), but with numerical coding for the treatment labels, this should not be a problem.
